ATTENUATION OF A RESISTIVE BRIDGE vs A HYBRID TRANSFORMER

Henry Hinrichs

PULSE INC.
Network Products Group

12220 World Trade Dr.
San Diego, CA 92128
Phone: (619) 674-8100
FAX: (619) 674-8262



RESISTIVE BRIDGE



HYBRID TRANSFORMER

TRANSMIT – MEDIA INSERTION LOSS SPICE MODEL

Transmit - Media attenuation, Bridge vs Hybrid Transformer

Date/Time run: 01/07/98 08:55:21

Temperature: 27.0





RESISTIVE BRIDGE



HYBRID TRANSFORMER

MEDIA – RECEIVE ATTENUATION SPICE MODEL


Media - Receive attenuation, Bridge vs Hybrid Transformer
Date/Time run: 01/07/98 09:02:04
Temperature: 27.0
(A) C:\MSIM\BRIDGER.DAT
-3.0
-3.2
-3.4
-3.6
-3.8
Hybrid Transformer
Resistive Bridge
100KHz
1.0MHz
10MHz
100MHz
1.0GHz
20*log10(v(15))
20*log10(v(3)-v(2))
Frequency
A1:(10.000M,-3.0103) A2:(10.000M,-3.5218) DIFF(A):(0.000,511.526m)

```
Transmit - Media Insertion Loss (S21), Bridge vs Hybrid Transformer

*     Synopsis:   The following represents a side by side comparison of
*                 the transmit - media attenuation of a resistive bridge
*                 verses a hybrid transformer.

*     Date: 01/07/98          H. Hinrichs
*                       Pulse Inc.
*                       12220 World Trade Dr.
*                       San Diego, CA, 92128
*                       Tel (619) 674-8208
*                       FAX (619) 674-8262

*     First the resistive bridge

IAC1  1     0     AC    .02985 ; Stimulus, bridge
RS1   1     0     51
R2    1     2     1000  ; receive negative input
R3    2     0     2000
R4    1     3     50    ; receive positive input
*     Transformer, 1:1 ratio with no high frequency parasitic losses
.param kval={1.0}
LP_Tran1    3     0     500uH
LS_Tran1    4     0     500uH
K_Tran1     LP_Tran1   LS_Tran1   {kval}
*     Media port
RL1   4     0     100   ; Characteristic impedance of cable

*     And now the Hybrid Transformer

IAC2  10    0     AC    .02828 ; Stimulus, hybrid transformer
RT1   10    0     100
*     Ideal hybrid transformer (again no parasitics)
*     Note: Coil inductances establish the following port
*           impedances:
*           Transmit = 100 ohms
*           Receive  = 100 ohms
*           Media    = 100 ohms
LT_Tran2    10    0     500uH
LM_Tran2    11    0     250uH
LD_Tran2    13    0     250uH
K_Tran2  LT_Tran2   LM_Tran2   LD_Tran2   {kval}
LR_Tran3    15    0     500uH
LM_Tran3    12    0     250uH
LD_Tran3    0     14    250uH
K_Tran3  LR_Tran3   LM_Tran3   LD_Tran3   {kval}
RR1   15    0     100 ; Receive port terminated in 100 ohms
RD1   13    14    100 ; Dummy 100 ohm resistive load
RM1   11    12    100 ; Media port terminated in 100 ohms

.probe
.ac   dec   100   100k  200meg
.end
```

```
Media - Receive attenuation, Bridge vs Hybrid Transformer

*     Synopsis:   The following represents a side by side comparison of
*                 the media - receive attenuation of a resistive bridge
*                 verses a hybrid transformer.

*     Date: 01/07/98          H. Hinrichs
*                       Pulse Inc.
*                       12220 World Trade Dr.
*                       San Diego, CA, 92128
*                       Tel (619) 674-8208
*                       FAX (619) 674-8262

*     First the resistive bridge

RS1   1     0     51
R2    1     2     1000  ; receive negative input
R3    2     0     2000
R4    1     3     50    ; receive positive input
*     Transformer, 1:1 ratio with no high frequency parasitic losses
.param kval={1.0}
LP_Tran1    3     0     500uH
LS_Tran1    4     0     500uH
K_Tran1     LP_Tran1   LS_Tran1   {kval}
*     Media port
RL1   4     5     100   ; Characteristic impedance of cable
*     Note: 1/2 of stimulus is dropped across far end source
VAC1  5     0     AC    2

*     And now the Hybrid Transformer

RT1   10    0     100
*     Ideal hybrid transformer (again no parasitics)
*     Note: Coil inductances establish the following port
*           impedances:

*           Transmit = 100 ohms
*           Receive  = 100 ohms
*           Media    = 100 ohms
LT_Tran2    10    0     500uH
LM_Tran2    11    0     250uH
LD_Tran2    13    0     250uH
K_Tran2  LT_Tran2   LM_Tran2   LD_Tran2   {kval}
LR_Tran3    15    0     500uH
LM_Tran3    12    0     250uH
LD_Tran3    0     14    250uH
K_Tran3  LR_Tran3   LM_Tran3   LD_Tran3   {kval}
RR1   15    0     100 ; Receive port terminated in 100 ohms
RD1   13    14    100 ; Dummy 100 ohm resistive load
RM1   11    16    100 ; Media port terminated in 100 ohms
*     Note: As with the bridge circuit 1/2 of stimulus is dropped across
*     far end source
VAC2  16    12    AC    2

.probe
.ac   dec   100   100k  200meg
.end
```